UCID-17277

Lawrence Livermore Laboratory

GRAY: A PROGRAM TO CALCULATE GRAY-BODY RADIATION HEAT-TRANSFER VIEW FACTORS FROM BLACK-BODY VIEW FACTORS

Robert L. Wong

June 14, 1976

MASTER



This is an informal report intended primarily for internal or limited external distribution. The opinions and conclusions stated are those of the author and may or may not be those of the laboratory
Prepared for U. S. Department of Energy under contract No. W-7405-Eng-48.

DISCLAIMER

PAGES ___i___ to ___ii___

WERE INTENTIONALLY

LEFT BLANK

CONTENTS

Abstract 1

Introduction 2

Section 1: Program Description 3

Section 2: Program Use 6

Section 3: Cylinder Example 11

References 19

Appendix: Program Listing 21

GRAY: A PROGRAM TO CALCULATE GRAY-BODY RADIATION HEAT-TRANSFER VIEW FACTORS FROM BLACK-BODY VIEW FACTORS

ABSTRACT

Program GRAY is written to perform the matrix manipulations necessary to convert black-body radiation heat-transfer view factors to gray-body view factors as required by thermal analyzer codes. The black-body view factors contain only geometric relationships. Program GRAY allows the effects of multiple gray-body reflections to be included. The resulting effective gray-body view factors can then be used with the corresponding fourth-power temperature differences to obtain the net radiative heat flux.

The program is written to accept a matrix input or the card image output generated by the black-body view factor program CNVUFAC. The resulting card image output generated by GRAY is in a form usable by the TRUMP thermal analyzer.

INTRODUCTION

To couple radiation heat-transfer phenomena into thermal analyzer computer programs (e.g., the LLL code TRUMP[1]), it is usually necessary to express the radiation interchange in terms of fourth-power temperature differences. The net rate of heat exchange, Q_{ij}, from surface i to surface j must be expressed in the form[2,3]

$$Q_{ij} = A_i \mathcal{F}_{ij} \sigma (T_i^4 - T_j^4) \quad (1)$$

where A_i is the area of surface i; σ is the Stefan-Boltzmann constant; and T is the temperature. The factor $\mathcal{F}_{ij}$ is called the effective gray-body view factor (sometimes also called shape or configuration factor). This factor accounts for all the radiation from surface i to surface j including that reflected from all surfaces in the enclosure.

The determination of $\mathcal{F}_{ij}$ requires a two-step procedure. First, the black-body view factor, F_{ij}, must be determined. This factor depends only on geometry and is the fraction of the area of surface i that is seen by surface j. The effects of gray-body emissivities and the resulting multiple reflections must be included. Generally, this requires some matrix manipulation.

Program GRAY is written to perform these matrix manipulations. A related program, CNVUFAC,[4] can be used to determine the necessary black-body view factors. The card image output from CNVUFAC is suitable for input into GRAY. In turn, GRAY produces card image output usable in TRUMP.[1]

SECTION 1

PROGRAM DESCRIPTION

Given a gray radiation enclosure, program GRAY calculates the gray-body view factors from the following:

1. The matrix of black-body view factors;
2. The surface areas; and
3. The surface emissivities.

The program uses equations developed by Hottel and Sarofim.[3]

Following the notation in Ref. 3, Hottel defines the total gray-body interchange area, $\overline{S_i S_j}$, as

$$\overline{S_i S_j} = A_i \mathcal{F}_{ij} = A_j \mathcal{F}_{ji} = \overline{S_j S_i} \tag{2}$$

It is more convenient to work with this area view-factor product than with the separate terms. Hottel's Eq. (3-31) gives $\overline{S_i S_j}$ as

$$\overline{S_i S_j} = \frac{A_i \varepsilon_i}{\rho_i} \left({}_jW_i / E_j - \delta_{ij}\, \varepsilon_j \right) \tag{3a}$$

$$= \frac{A_j \varepsilon_j}{\rho_j} \left({}_iW_j / E_i - \delta_{ij} \varepsilon_i \right) \tag{3b}$$

where ε is the emissivity, ρ is the reflectivity, and δ_{ij} is the Kronecker delta. Since it is assumed that no radiation is transmitted,

$$\rho = 1 - \varepsilon \; . \tag{4}$$

W is the total radiative flux leaving the surface or the radiosity. This includes the flux that is emitted and the flux that is reflected. ${}_jW_i$ is the total flux leaving surface i due only to emission from surface j. E_j is the hemispherical black-body emissive power of surface j; therefore,

$$E_j = \sigma T_j^4 \tag{5}$$

The radiosity ${}_jW_i$ is given by Hottel's Eq. (3-36) as

$$ {}_jW_i = \frac{A_j \varepsilon_j}{\rho_j} E_j \left(- \frac{D'_{ij}}{D} \right) \tag{6} $$

In Eq. (6), D is the determinant of the transfer matrix given in Hottel's radiative interchange Eq. (3-25):

$$ \underbrace{\begin{bmatrix} \left(\overline{S_1S_1} - \frac{A_1}{\rho_1}\right) & \overline{S_1S_2} & \overline{S_1S_3} & \cdots \\ \overline{S_1S_2} & \left(\overline{S_2S_2} - \frac{A_2}{\rho_2}\right) & \overline{S_2S_3} & \cdots \\ \overline{S_1S_3} & \overline{S_2S_3} & \left(\overline{S_3S_3} - \frac{A_3}{\rho_3}\right) & \cdots \\ \vdots & \vdots & \vdots & \end{bmatrix}}_{\text{Transfer Matrix} = D} \bullet \underbrace{\begin{bmatrix} W_1 \\ W_2 \\ W_3 \\ \vdots \end{bmatrix}}_{\text{Response Vector}} \underbrace{\begin{bmatrix} - \frac{A_1 \varepsilon_1}{\rho_1} E_1 \\ - \frac{A_2 \varepsilon_2}{\rho_2} E_2 \\ - \frac{A_3 \varepsilon_3}{\rho_3} E_3 \\ \vdots \end{bmatrix}}_{\text{Excitation Vector}} \quad i \tag{7} $$

D'_{ij} is the cofactor or signed minor of column i and row j of D.

To enable the program to consider black surfaces ($\rho_i = 0$), each row i in Eq. (7) is multiplied by ρ_i. The appropriate change in the definitions of ${}_jD_i$ and D are allowed for Eq. (6), which becomes

$$ {}_jW_i = A_j \varepsilon_j E_j \left(- \frac{D'_{ij}}{D} \right) \tag{8} $$

Even with this modification, additional care must be used to evaluate black surfaces. If one of the two surfaces i and j considered is black, it is noted that

$$ \overline{S_iS_j} = \overline{S_jS_i} \quad , $$

and the appropriate Eq. (3a) or (3b) is used. If surface j is black, Eq. (3a) is used, and if surface i is black, Eq. (3b) is used. If both surfaces i and j are black, Hottel's Eq. (3-53) is used:

$$\overline{S_j S_i} = \overline{s_j s_i} + \frac{1}{E_j} \sum s_k s_i \, {}_j W_k \tag{9}$$

where $s_j s_i$ is the black-body total interchange area.

$$s_i s_j = A_i F_{ij} = A_j F_{ji} = s_j s_i \tag{10}$$

To evaluate the view factors, F_{ij} and F_{ji}, Eq. (2) is used.

The term (D'_{ij}/D) in Eq. (8) is determined from the inverse matrix using the inversion subroutine MATINV.[5] It must be noted that the inverse matrix is the transpose of the cofactor matrix divided by the determinant.

SECTION 2

PROGRAM USE

2.1 OBTAINING THE PROGRAM

In the ELF system, the executable code is stored as

.980012:OBJECT:GRAY

If the Fortran code is desired, it is stored as

.980012:FORTRAN:FGRAY

2.2 PROGRAM EXECUTION

GRAY is executed from the teletype using one of the three options shown in Fig. 1. The name of the input file is requested if it is not included on the execution line. If there is no input file name given, GRAY assumes the name is GRYIN.

The three output file names are constructed from the input file name by appending the first three characters to the front of OUT for the output file, and appending them to the front of BLK5 and BLK6 for the two TRUMP[1] card image input files.

2.3 PROGRAM INPUT

Figure 2 shows the program input format. Card 1 contains the problem title. Card 2 is the program control card. Card 3 defines the node numbers. Card 4 contains the surface emissivities. Cards, 5(a) and 5(b) allow a choice of two input formats for the black-body view factor data. Card 5(a) is used for a matrix input (KDATA = 0). Card 5(b) is used for a tabular input (KDATA = 1) that is in the card image output format of program CNVUFAC.

Card 1, Title Card, 8A10

Problem title.

```
GRAY / .2 .22

TYPE INPUT FILE NAME.
FIRST 3 CHARACTERS ARE APPENDED TO OUTPUT FILE NAMES.


OUTPUT FILE NAMES ARE:
OUTPUT= GRYOUT
TRUMP BLOCK 5= GRYBLK5
TRUMP BLOCK 6= GRYBLK6

 ALL DONE


GRAY / .2 .22

TYPE INPUT FILE NAME.
FIRST 3 CHARACTERS ARE APPENDED TO OUTPUT FILE NAMES.
GRYIN

OUTPUT FILE NAMES ARE:
OUTPUT= GRYOUT
TRUMP BLOCK 5= GRYBLK5
TRUMP BLOCK 6= GRYBLK6

 ALL DONE

GRAY GRYIN / .2 .22

OUTPUT FILE NAMES ARE:
OUTPUT= GRYOUT
TRUMP BLOCK 5= GRYBLK5
TRUMP BLOCK 6= GRYBLK6

 ALL DONE
```

FIG. 1. Teletype execution of GRAY.



FIG. 2. Program GRAY input format.

Card 2, Program Control, 3I10

N = Number of nodes considered. The program is currently dimensioned for a maximum of 20.

KDATA = Control for the format of the black-body view factor data input.

KDATA = 0 if the black-body interchange areas, SSB, are input in matrix form [Card 5(a)].

KDATA = 1 if these data are input in tabular form [Card 5(b)] as generated by CNVUFAC.

NEXT = 0 if this is the last problem.

NEXT = 1 if another problem follows.

Card 3, Node Number Identification, 8I10

NODE = Node identification number. It is also used as a flag to identify whether the node is an internal node (in the sense of the TRUMP[1] code) or an external node to which a boundary condition is to be applied. NODE is zero or positive for an internal node (TRUMP input BLOCK 5). It is negative for an external node (TRUMP input BLOCK 6).

Card 4, Node Emissivities, 8E10.3

E = Node emissivity. E = 1.0 for a black-body node. E = 0.0 for a radiatively adiabatic node.[3] This is a node at which the leaving flux equals the incident flux, an assumption often made for refractory walls. The temperature of a radiatively adiabatic node does not enter into the solution of the interchange problem so it does not have to be included in the TRUMP input.

Card 5, Black-Body Interchange Areas, 8E10.3

SSB = Black-body interchange area ($\overline{s_i s_j}$ in Hottel). This is the area view-factor product. Note that the matrix is symmetric:

$$SSB(I,J) = SSB(J,I) \; ; \qquad (11)$$

and that the area of node I, A(I), is determined in the program by summing over all the interchange areas in row I:

$$A(I) = \sum_{J=1}^{N} SSB(I,J). \tag{12}$$

As a result of Eq. (12), all the interchange areas for a complete enclosure must be defined.

In the input matrix format (KDATA = 0), only the lower triangular matrix is input (i.e., $J \leq I$). To ensure the symmetry of the matrix, Eq. (11) is used to determine the remainder of the matrix.

In the tabular input format (KDATA = 1), the input is in the output card image format of program CNVUFAC.[4] All the nodes appearing on card 5(b) must be listed on the node number identification card 3 [external nodes need not be made negative in card 5(b)]. Note that the order of the two node numbers listed with an interchange area makes no difference. If any node-to-node interchanges are listed more than once, the interchange areas are averaged. All interchanges that are omitted are assumed to be zero.

SECTION 3

CYLINDER EXAMPLE

The cylinder example is used to evaluate the gray-body interchange areas for the inside of the cylinder shown in Fig. 3. This figure is taken from Ref. 4 where the black-body interchange areas are determined using CNVUFAC. The card image output from CNVUFAC is used to form the input shown in Fig. 4.

All surfaces have an emissivity of 0.3, except nodes 1 and 21 at the centers of the ends. (The node numbers are indicated as circled values in Fig. 3). These two center end nodes are holes or black surfaces to which boundary conditions are to be applied when solving the problem using TRUMP[1] (external nodes). Several interchange areas inputs are duplicated (since the order in which the nodes are listed makes no difference). However, this is acceptable since, as mentioned in Section 2, multiple inputs are averaged.

Figure 5 shows the input data. The black-body interchange areas are shown in matrix form. Note that this is the format in which these data would have been input if the matrix input option had been selected.

Figure 6 shows the matrix of gray-effective interchange areas and the matrix of view factors. Also shown are the node areas A(I), used as obtained by adding the values of SSB(I,J) over all elements J of row I.

Figures 7 and 8 show the card files written for TRUMP BLOCK 5 and TRUMP BLOCK 6, respectively. The BLOCK 5 file contains all the interchanges between internal nodes. The BLOCK 6 file contains all the interchanges between internal and external nodes. Interchanges between external nodes are not included since they do not enter into the TRUMP calculation. Note that if any radiatively adiabatic nodes would have been present (E = 0.0), the interchanges with these nodes also would not have appeared.

Figures 9 and 10 show, respectively, the actual TRUMP BLOCK 5 and TRUMP BLOCK 6 files. These cards can be used directly in the TRUMP input if the nodes are defined in TRUMP BLOCK 4 as zero volume surface nodes. Note that the areas are set to 1.0 since the interchange area instead of the view factor is used.



FIG. 3. Cylinder example.

```
GRAY CYLINDER EXAMPLE: CENTER OF ENDS ARE BLACK AND EXTERIOR NODES.
         7          1          0
        -1          2         11         12         13        -21         22
1.0          0.3       0.3        0.3        0.3        1.0        0.3
         1         11 4.164E+01
         1         12 2.178E+01
         1         13 7.430E+00
         2         11 1.556E+02
         2         12 4.449E+01
         2         13 1.771E+01
        21         13 4.164E+01
        21         12 2.178E+01
        21         11 7.430E+00
        22         13 1.556E+02
        22         12 4.449E+01
        22         11 1.771E+01
        11         11 2.398E+02
        11         12 1.281E+02
        11         13 4.099E+01
        12         11 1.281E+02
        12         12 2.398E+02
        12         13 1.281E+02
        13         11 4.099E+01
        13         12 1.281E+02
        13         13 2.398E+02
         1         21 2.072E+00
         1         22 5.620E+00
         2         21 5.620E+00
         2         22 1.552E+01
         0          0      .0E+00
```

FIG. 4. Input for cylinder example.

```
                           +++++ GRAY BODY RADIATION HEAT TRANSFER SHAPE FACTOR PROGRAM +++++
GRAY CYLINDER EXAMPLE: CENTER OF ENDS ARE BLACK AND EXTERIOR NODES.

+++++ INPUT +++++

    N        KDATA       NEXT
         7          1          0

 NODE(I)
        -1          2         11         12         13        -21         22

  E(I)
1.000E+00 3.000E-01 3.000E-01 3.000E-01 3.000E-01 1.000E+00 3.000E-01

SSB(I,J)
   .0E+00
   .0E+00    .0E+00
4.164E+01 1.556E+02 2.398E+02
2.178E+01 4.449E+01 1.281E+02 2.398E+02
7.430E+00 1.771E+01 4.099E+01 1.281E+02 2.398E+02
2.072E+00 5.620E+00 7.430E+00 2.178E+01 4.164E+01    .0E+00
5.620E+00 1.552E+01 1.771E+01 4.449E+01 1.556E+02    .0E+00    .0E+00
```

FIG. 5. Output from cylinder example: input data.

1

+++++ GRAY BODY RADIATION HEAT TRANSFER SHAPE FACTOR PROGRAM +++++

GRAY CYLINDER EXAMPLE: CENTER OF ENDS ARE BLACK AND EXTERIOR NODES.

+++++ MATRIX OF GRAY TOTAL INTERCHANGE AREAS, SS(I,J)=A(I)*F(I,J)=Q(I,J)/(SIGMA*(T(I)**4-T(J)**4))=SS(J,I) +++++

	1	2	11	12	13	21	22
1	5.960E+00	5.720E+00	2.519E+01	1.807E+01	1.185E+01	6.363E+00	5.381E+00
2	5.720E+00	5.543E+00	2.629E+01	1.427E+01	9.809E+00	5.381E+00	4.569E+00
11	2.519E+01	2.629E+01	5.515E+01	3.742E+01	2.366E+01	1.185E+01	9.809E+00
12	1.807E+01	1.427E+01	3.742E+01	4.903E+01	3.742E+01	1.807E+01	1.427E+01
13	1.185E+01	9.809E+00	2.366E-01	3.742E+01	5.515E+01	2.519E+01	2.629E+01
21	6.363E+00	5.381E+00	1.185E-01	1.807E+01	2.519E+01	5.960E+00	5.720E+00
22	5.381E+00	4.569E+00	9.809E-00	1.427E+01	2.629E+01	5.720E+00	5.643E+00

+++++ MATRIX OF GRAY SHAPE FACTORS, F(I,J)=SS(I,J)/A(I) +++++

	1	2	11	12	13	21	22
1	7.588E-02	7.283E-02	3.208E-01	2.301E-01	1.509E-01	8.102E-02	6.852E-02
2	2.394E-02	2.362E-02	1.100E-01	5.974E-02	4.105E-02	2.252E-02	1.912E-02
11	3.991E-02	4.164E-02	8.737E-02	5.928E-02	3.748E-02	1.878E-02	1.554E-02
12	2.875E-02	2.271E-02	5.954E-02	7.800E-02	5.954E-02	2.875E-02	2.271E-02
13	1.878E-02	1.554E-02	3.748E-02	5.928E-02	8.737E-02	3.991E-02	4.164E-02
21	8.102E-02	6.852E-02	1.509E-01	2.301E-01	3.208E-01	7.588E-02	7.283E-02
22	2.252E-02	1.912E-02	4.105E-02	5.974E-02	1.100E-01	2.394E-02	2.362E-02

+++++ AREAS, A(I), FROM THE SUM OF SSB(I,J) OVER J +++++

1	2	11	12	13	21	22
7.854E+01	2.389E-02	6.313E+02	6.285E+02	6.313E+02	7.854E+01	2.389E+02

FIG. 6. Output from cylinder example: matrix of gray effective interchange areas and matrix of view factors.

1

+++++ GRAY BODY RADIATION HEAT TRANSFER SHAPE FACTOR PROGRAM +++++

GRAY CYLINDER EXAMFLE: CENTER OF ENDS ARE BLACK AND EXTERIOR NODES.

+++++ TRUMP BLOCK 5 CARDS (INTERNAL RADIATION CONNECTIONS) WRITTEN INTO FILE JGRBLK5 +++++

NOD1	NOD2	DEL1	DEL2	DLONG	DRAD	HINT	RINT
BLOCK 5							
11	2	.0E+00	.0E+00	1.000E+00	1.000E+00	1.000E-24	2.629E+01
12	2	.0E+00	.0E+00	1.000E+00	1.000E+00	1.000E-24	1.427E+01
12	11	.0E+00	.0E+00	1.000E+00	1.000E+00	1.000E-24	3.742E+01
13	2	.0E+00	.0E+00	1.000E+00	1.000E+00	1.000E-24	9.809E+00
13	11	.0E+00	.0E+00	1.000E+00	1.000E+00	1.000E-24	2.366E+01
13	12	.0E+00	.0E+00	1.000E+00	1.000E+00	1.000E-24	3.742E+01
22	2	.0E+00	.0E+00	1.000E+00	1.000E+00	1.000E-24	4.569E+00
22	11	.0E+00	.0E+00	1.000E+00	1.000E+00	1.000E-24	9.809E+00
22	12	.0E+00	.0E+00	1.000E+00	1.000E+00	1.000E-24	1.427E+01
22	13	.0E+00	.[illegible]E+00	1.000E+00	1.000E+00	1.000E-24	2.629E+01
0							

FIG. 7. Output from cylinder example: card files written for TRUMP BLOCK 5.

```
1
                                      ++++- GRAY BODY RADIATION HEAT TRANSFER SHAPE FACTOR PROGRAM +++++

 GRAY CYLINDER EXAMPLE: CENTER OF ENDS ARE BLACK AND EXTERIOR NODES.

 +-+++ TRUMP BLOCK 6 CARDS (EXTERNAL RADIATION CONNECTIONS) WRITTEN INTO FILE JGRBLK6 +++++

NODS NODSB                          DLONG      DRAD       HSURE      RSURE      POWER
BLOCK 6
    2    1                      1.000E+00 1.000E+00    .0E+00 5.720E+00    .0E+00
    2   21                      1.000E+00 1.000E+00    .0E+00 5.381E+00    .0E+00
   11    1                      1.000E+00 1.000E+00    .0E+00 2.519E+01    .0E+00
   11   21                      1.000E+00 1.000E+00    .0E+00 1.185E+01    .0E+00
   12    1                      1.000E+00 1.000E+00    .0E+00 1.807E+01    .0E+00
   12   21                      1.000E+00 1.000E+00    .0E+00 1.807E+01    .0E+00
   13    1                      1.000E+00 1.000E+00    .0E+00 1.185E+01    .0E+00
   13   21                      1.000E+00 1.000E+00    .0E+00 2.519E+01    .0E+00
   22    1                      1.000E+00 1.000E+00    .0E+00 5.381E+00    .0E+00
   22   21                      1.000E+00 1.000E+00    .0E+00 5.720E+00    .0E+00
    0
```

FIG. 8. Output from cylinder example: card files written for TRUMP BLOCK 6.

```
 GRAY CYLINDER EXAMPLE: CENTER OF ENDS ARE BLACK AND EXTERIOR NODES.
BLOCK 6
    2    1                    1.000E+00 1.000E+00    .0E+00 5.720E+00    .0E+00
    2   21                    1.000E+00 1.000E+00    .0E+00 5.381E+00    .0E+00
   11    1                    1.000E+00 1.000E+00    .0E+00 2.519E+01    .0E+00
   11   21                    1.000E+00 1.000E+00    .0E+00 1.185E+01    .0E+00
   12    1                    1.000E+00 1.000E+00    .0E+00 1.807E+01    .0E+00
   12   21                    1.000E+00 1.000E+00    .0E+00 1.807E+01    .0E+00
   13    1                    1.000E+00 1.000E+00    .0E+00 1.185E+01    .0E+00
   13   21                    1.000E+00 1.000E+00    .0E+00 2.519E+01    .0E+00
   22    1                    1.000E+00 1.000E+00    .0E+00 5.381E+00    .0E+00
   22   21                    1.000E+00 1.000E+00    .0E+00 5.720E+00    .0E+00
    0
```

FIG. 9. Cylinder example: TRUMP BLOCK 5 file.

```
 GRAY CYLINDER EXAMPLE: CENTER OF ENDS ARE BLACK AND EXTERIOR NODES.
BLOCK 5
   11     2                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 2.629E+01
   12     2                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 1.427E+01
   12    11                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 3.742E+01
   13     2                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 9.809E+00
   13    11                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 2.366E+01
   13    12                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 3.742E+01
   22     2                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 4.569E+00
   22    11                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 9.809E+00
   22    12                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 1.427E+01
   22    13                .0E+00     .0E+00 1.000E+00 1.000E+00 1.000E-24 2.629E+01
    0
```

FIG. 10. Cylinder example: TRUMP BLOCK 6 file.

REFERENCES

1. A. L. Edwards, TRUMP: A Computer Program for Transient and Steady-State Temperature Distributions in Multidimensional Systems, Lawrence Livermore Laboratory, Livermore, California, UCRL-14754, Rev. 3 (September 1, 1972).
2. R. Siegel and J. R. Howell, Thermal Radiation Heat Transfer (McGraw-Hill Book Company, Inc., New York, 1972).
3. H. C. Hottel and A. F. Sarofim, Radiative Transfer (McGraw-Hill Book Company, Inc., New York, 1967).
4. R. L. Wong, User's Manual for CNVUFAC, The General Dynamics Radiation Heat-Transfer View-Factor Program, Lawrence Livermore Laboratory, Livermore, California, UCID-17275, June 25, 1976.
5. B. S. Garbow, Matrix Inversion With Accompanying Solution of Linear Equations, Lawrence Livermore Laboratory, Livermore, California, CIC Rept. F1.2-002 (April 8, 1963).

R.M.

THIS PAGE

WAS INTENTIONALLY

LEFT BLANK

APPENDIX

PROGRAM LISTING

```
      PROGRAM GRAY(TAPE2,TAPE3,TAPE4,TAPE5)
C
C---THIS PROGRAM SOLVES FOR THE GRAY BODY RADIATION HEAT TRANSFER
C   INTERCHANGE AREA, SS, AND SHAPE FACTOR, F, MATRICIES GIVEN:
C   1. THE BLACK BODY GEOMETRIC INTERCHANGE AREA MATRIX, SSB,
C   2. NODE EMISSIVITIES, E.
C   SSB CAN BE EITHER INPUT AS A LOWER TRIANGULAR MATRIX OR AS A LIST.
C   THIS LIST FORMAT IS CONSISTANT WITH THE PUNCH OUTPUT FROM CNVUFAC
C   (THE HEADING CARDS MUST BE REMOVED).
C   THESE GRAY BODY SHAPE FACTORS ARE SUITABLE FOR USE IN THE TRUMP
C   HEAT TRANSFER CODE.  TRUMP BLOCKS 5 AND 6 (INTERNAL AND EXTERNAL
C   RADIATION CONNECTIONS) ARE BE GENERATED.
C   THE RADIATION HEAT FLUX IS GIVEN BY:
C            Q(I,J)=SS(I,J)*SIGMA*(T(I)**4-T(J)**4)
C           SS(I,J)=A(I)*F(I,J)=SS(J,I)
C   NOTE THAT E=0.0 FOR RADIATIVELY ADIABATIC NODES.
C   REFERENCE:  H.C. HOTTEL AND A.F. SAROFIM, RADIATIVE HEAT TRANSFER,
C   MCGRAW-HILL, 1967, PP 72-121.
C   R.L. WONG, LAWRENCE LIVERMORE LABORATORY, MAY 1976.
C
C---THE NAMES OF THE OUTPUT FILES ARE OBTAINED BY APPENDING THE
C   FIRST 3 CHARACTERS (OR ALL CHARACTERS IF 3 OR LESS) OF
C   THE INPUT FILE AS FOLLOWS:  IF INPUT FILE IS XXXIN,
C   THE OUTPUT FILE IS XXXOUT,
C   THE TRUMP BLOCK 5 OUTPUT FILE IS XXXBLK5,
C   THE TRUMP BLOCK 6 OUTPUT FILE IS XXXBLK6,
C   THE DEFAULT FOR XXX IS GRY.
C
C---NOTE THAT DIMENSIONING IS CURRENTLY FOR 20 NODES.  THE DIMENSIONS IN
C   SUBROUTINES MATINV AND ORDER MUST BE INCREASED IF THE DIMENSIONS IN
C   THE MAIN PROGRAM ARE INCREASED TO HANDLE MORE THAN 20 NODES.
      COMMON HEAD(8),A(20),E(20),SSB(20,20),SS(20,20),F(20,20),
     1 RHO(20),B(20),WE(20,20),KE(20),NODE(20),KEXT(20),FNAME(3),N
      DATA DEL1,DEL2/2*0.0/,DLONG,DRAD/2*1.0/,HINT/1.0E-24/,
     1 HSURE,POWER/2*0.0/,NEND/0/,ITY/59/
C
C---FORMAT STATEMENTS.
    1 FORMAT(1H1)
    2 FORMAT(8A10)
    3 FORMAT(8I10)
    4 FORMAT(8E10.3)
    5 FORMAT(27X,66H+++++ GRAY BODY RADIATION HEAT TRANSFER SHAPE FACTOR
     1 PROGRAM +++++/1X)
    6 FORMAT(/18H +++++ INPUT +++++)
    7 FORMAT(/5X,1HN,4X,3X,5HKDATA,2X,3X,4HNEXT)
    8 FORMAT(/2X,7HNODE(I))
   10 FORMAT(/3X,4HE(I))
   11 FORMAT(/1X,8HSSB(I,J))
   12 FORMAT(/113H +++++ MATRIX OF GRAY TOTAL INTERCHANGE AREAS, SS(I,J)
     1=A(I)*F(I,J)=Q(I,J)/(SIGMA*(T(I)**4-T(J)**4))=SS(J,I) +++++)
   13 FORMAT(/62H +++++ MATRIX OF GRAY SHAPE FACTORS, F(I,J)=SS(I,J)/A(I
     1) +++++)
   14 FORMAT(7X,10I10)
   15 FORMAT(I10,10E10.3/(10X,10E10.3))
   16 FORMAT(/57H +++++ AREAS, A(I), FROM THE SUM OF SSB(I,J) OVER J +++
     1++)
   17 FORMAT(10X,10E10.3)
   18 FORMAT(/78H +++++ TRUMP BLOCK 5 CARDS (INTERNAL RADIATION CONNECTI
     1ONS) WRITTEN INTO FILE ,A7,6H +++++)
   19 FORMAT(/78H +++++ TRUMP BLOCK 6 CARDS (EXTERNAL RADIATION CONNECTI
```

```
     1ONS) WRITTEN INTO FILE ,A7,6H +++++)
   20 FORMAT(/1X,4HNOD1,1X,4HNOD2,10X,3X,4HDEL1,3X,3X,4HDEL2,3X,3X,
     1 5HDLONG,2X,3X,4HDRAD,3X,3X,4HHINT,3X,3X,4HRINT)
   21 FORMAT(/4HNODS,1X,5HNODSB,20X,3X,5HDLONG,2X,3X,4HDRAD,3X,3X,
     1 5HHSURE,2X,3X,5HRSURE,2X,3X,5HPOWER)
   22 FORMAT(7HBLOCK 5)
   23 FORMAT(7HBLOCK 6)
   24 FORMAT(2I5,10X,6E10.3)
   25 FORMAT(2I5,20X,5E10.3)
   26 FORMAT(I5)
   30 FORMAT(/21HTYPE INPUT FILE NAME./
     1 53HFIRST 3 CHARACTERS ARE APPENDED TO OUTPUT FILE NAMES.)
   32 FORMAT(/22HOUTPUT FILE NAMES ARE:/8HOUTPUT= ,A6/
     1 15HTRUMP BLOCK 5= ,A7/15HTRUMP BLOCK 6= ,A7)
   33 FORMAT(/17HFLOE ERROR NUMBER,I5,30H FOR OPEN INSTRUCTION OF FILE ,
     1 A10)
C
C---SET UP FILES.
C    READ INPUT FILE NAME FROM TELETYPE.
      CALL GOB(1603B,IER,700B,KKK)
      IF(IER.EQ.0) GO TO 50
      WRITE(ITY,30)
C
C    FNAMEI=INPUT FILE NAME.  THE FIRST 3 CHARACTERS (OR ALL
C    CHARACTERS IF 3 OR LESS) ARE APPENDED TO THE BEGINNING OF ALL
C    OUTPUT FILES.  THE DEFAULT FOR FNAMEI IS GRYIN.
   50 READ(ITY,2) FNAMEI
C
C    CHECK OF FNAMEI IS BLANK OR CONTAINS LESS THAN 3 CHARACTERS.
      IF(FNAMEI.EQ.1H ) FNAMEI=5HGRYIN
      IS=0
      IBYTE=LBYT(FNAMEI,48,6)
      IF(IBYTE.NE.1R ) GO TO 52
      IS=12
      GO TO 53
   52 IBYTE=LBYT(FNAMEI,42,8)
      IF(IBYTE.NE.1R ) GO TO 53
      IS=6
   53 DO 55 L=1,3
      FNAME(L)=FNAMEI
   55 CONTINUE
      CALL SBYT(FNAME(1),IS,42,7ROUT      )
      CALL SBYT(FNAME(2),IS,42,7RBLK5     )
      CALL SBYT(FNAME(3),IS,42,7RBLK6     )
      WRITE(ITY,32) FNAME
C
C    OPEN INPUT FILE.
      IE=-2
      CALL OPENFN(FNAMEI,2,LGIN,IE)
      IF(IE.EQ.0) GO TO 58
      WRITE(ITY,33) IE,FNAMEI
      CALL EXIT
C
C    OPEN AND CREATE OUTPUT FILES.
   58 DO 60 L=1,3
      IE=-5
      LG=5000B
      NT=L+2
      CALL OPENFN(FNAME(L),NT,LG,IE)
      IF(IE.EQ.0) GO TO 60
```

```
      WRITE(ITY,33) IE,FNAME(L)
      CALL EXIT
   60 CONTINUE
C
C---INPUT.
C
C   HEAD=HEADING INPUT CARD.
  100 READ(2,2) HEAD
      WRITE(3,1)
      WRITE(3,5)
      WRITE(3,2) HEAD
C
C   N=NUMBER OF SURFACES.
C   KDATA=0 IF INTERCHANGE AREAS ARE INPUT AS A MATRIX, =1 IF THEY ARE
C   INPUT AS A LIST (AS GENERATED BY CNVUFAC) TO BE ORGANIZED BY THE
C   PROGRAM INTO A MATRIX.
C   NEXT=1 IF ANOTHER DATA SET FOLLOWS, =0 IF THIS IS LAST DATA SET.
      READ(2,3) N,KDATA,NEXT
      WRITE(3,6)
      WRITE(3,7)
      WRITE(3,3) N,KDATA,NEXT
C
C   NODE=NODE IDENTIFICATION NUMBER.  MAKE NODE(I) NEGATIVE IF IN TRUMP
C   BLOCK 6, NODE(I) IS AN EXTERIOR NODE.  NODE(I)=0 IS ASSUMED TO BE
C   AN INTERIOR NODE.  THIS SIGN CONVENTION ONLY EFFECTS THE GENERATION
C   OF THE BLOCK 5 AND 6 CARDS.
      READ(2,3) (NODE(I), I=1,N)
      WRITE(3,8)
      WRITE(3,3) (NODE(I), I=1,N)
      DO 105 I=1,N
      KEXT(I)=0
      IF(NODE(I).GE.0) GO TO 105
      KEXT(I)=1
      NODE(I)=IABS(NODE(I))
  105 CONTINUE
C
C   E=NODE EMISSIVITIES (SET E=0.0 FOR RADIATIVELY ADIABATIC NODES).
      READ(2,4) (E(I), I=1,N)
      WRITE(3,10)
      WRITE(3,4) (E(I), I=1,N)
      WRITE(3,11)
      IF(KDATA.EQ.0) GO TO 106
C
C   IF INTERCHANGE AREAS, SSB, ARE IN LIST FORM (AS GENERATED BY PROGRAM
C   CNVUFAC), CALL SUBROUTINE ORDER TO READ AND ORGANIZE IN MATRIX FORM.
      CALL ORDER
      GO TO 109
C
C   INPUT IF INTERCHANGE AREAS, SSB, ARE IN MATRIX FORM.
C   SSB=A*FB=BLACK BODY GEOMETRIC TOTAL INTERCHANGE AREA.  THE MATRIX
C   IS SYMMETRIC.  ONLY THE LOWER TRIANGULAR MATRIX IS READ IN
C   (SSB(I,J), J.LE.I).  THE REST IS GENERATED.  THE SUM OF SSB(I,J)0
C   OVER J IS USED TO DETERMINE THE AREA, A(I).
  106 DO 107 I=1,N
      READ(2,4) (SSB(I,J), J=1,I)
  107 CONTINUE
  109 DO 110 I=1,N
      WRITE(3,4) (SSB(I,J), J=1,I)
      DO 110 J=1,I
      SSB(J,I)=SSB(I,J)
```

```
  110 CONTINUE
C
C    CALCULATE AREA, A(I), BY SUMMING SSB(I,J) OVER J.
      DO 115 I=1,N
      A(I)=0.0
      DO 115 J=1,N
      A(I)=A(I)+SSB(I,J)
  115 CONTINUE
C
C---BEGIN CALCULATION.
      DO 130 I=1,N
      RHO(I)=1.0-E(I)
C
C    CHECK FOR BLACK BODY.
      KE(I)=0
      IF(RHO(I).LT.1.0E-6) KE(I)=1
C
C    GENERATE TRANSFER MATRIX (MULTIPLY HOTTEL AND SAROFIM EQN.(3-25) BY
C    RHO(I) TO ALLOW FOR BLACK SURFACES).
      DO 120 J=1,N
      SS(I,J)=SSB(I,J)*RHO(I)
  120 CONTINUE
      SS(I,I)=SS(I,I)-A(I)
  130 CONTINUE
C
C    INVERT TRANSFER MATRIX.
      M=0
      CALL MATINV(SS,N,B,M,DETERM)
C
C    GENERATE TOTAL INTERCHANGE AREA MATRIX, SS, (HOTTEL AND SAROFIM,
C    EQN.(3-31,36,39,53) RECALLING EQN.(3-25) WAS MULTIPLIED BY RHO(I).
C
C    GENERATE W/E MATRIX, WE.
      DO 140 I=1,N
      DO 140 J=1,N
      WE(I,J)=-A(I)*E(I)*SS(J,I)
  140 CONTINUE
      DO 200 I=1,N
C
C    GENERATE DIAGONAL ELEMENTS OF SS MATRIX.
      IF(KE(I).EQ.1) GO TO 150
      SS(I,I)=A(I)*E(I)/RHO(I)*(WE(I,I)-E(I))
      GO TO 160
  150 SUM=0.0
      DO 155 K=1,N
      IF(K.EQ.I) GO TO 155
      SUM=SUM+SSB(K,I)*WE(I,K)
  155 CONTINUE
      SS(I,I)=SSB(I,I)+SUM
C
C    GENERATE OFF-DIAGONAL ELEMENTS OF SS MATRIX.
  160 IF(I.EQ.1) GO TO 200
      I1=I-1
      DO 200 J=1,I1
      IF(KE(I).EQ.1) GO TO 170
      SS(I,J)=A(I)*E(I)/RHO(I)*WE(J,I)
      GO TO 190
  170 IF(KE(J).EQ.1) GO TO 180
      SS(I,J)=A(J)*E(J)/RHO(J)*WE(I,J)
      GO TO 190
```

```
  180 SUM=0.0
      DO 185 K=1,N
      IF(K.EQ.I) GO TO 185
      SUM=SUM+SSB(K,J)*WE(I,K)
  185 CONTINUE
      SS(I,J)=SSB(I,J)+SUM
  190 SS(J,I)=SS(I,J)
  200 CONTINUE
C
C---OUTPUT.
C
C   OUTPUT TOTAL INTERCHANGE AREA MATRIX, SS.
      WRITE(3,1)
      WRITE(3,5)
      WRITE(3,2) HEAD
      WRITE(3,12)
      WRITE(3,14) (NODE(J), J=1,N)
      DO 300 I=1,N
      WRITE(3,15) NODE(I),(SS(I,J), J=1,N)
  300 CONTINUE
C
C   GENERATE AND OUTPUT SHAPE FACTOR MATRIX, F.
      DO 310 I=1,N
      DO 310 J=1,N
      F(I,J)=SS(I,J)/A(I)
  310 CONTINUE
      WRITE(3,13)
      WRITE(3,14) (NODE(J), J=1,N)
      DO 320 I=1,N
      WRITE(3,15) NODE(I),(F(I,J), J=1,N)
  320 CONTINUE
      WRITE(3,16)
      WRITE(3,14) (NODE(J), J=1,N)
      WRITE(3,17) (A(J), J=1,N)
C
C---WRITE FILES CONTAINING TRUMP BLOCKS 5 AND 6 CARDS.  THEY ARE NOT
C   AUTOMATICALLY PUNCHED.
C
C   TRUMP BLOCK 5 CARDS (INTERNAL RADIATION CONNECTIONS WITH
C   HINT=1.0E-24).
      WRITE(3,1)
      WRITE(3,5)
      WRITE(3,2) HEAD
      WRITE(3,18) FNAME(2)
      WRITE(3,20)
      WRITE(4,2) HEAD
      WRITE(3,22)
      WRITE(4,22)
      DO 350 I=2,N
      I1=I-1
      DO 350 J=1,I1
      IF((KEXT(I).EQ.1).OR.(KEXT(J).EQ.1)) GO TO 350
      IF(SS(I,J).LT.1.0E-20) GO TO 350
      WRITE(3,24) NODE(I),NODE(J),DEL1,DEL2,DLONG,DRAD,HINT,SS(I,J)
      WRITE(4,24) NODE(I),NODE(J),DEL1,DEL2,DLONG,DRAD,HINT,SS(I,J)
  350 CONTINUE
      WRITE(3,26) NEND
      WRITE(4,26) NEND
C
C   TRUMP BLOCK 6 CARDS (EXTERNAL RADIATION CONNECTIONS WITH HSURE=0.0).
```

```
      WRITE(3,1)
      WRITE(3,5)
      WRITE(3,2) HEAD
      WRITE(3,19) FNAME(3)
      WRITE(3,21)
      WRITE(5,2) HEAD
      WRITE(3,23)
      WRITE(5,23)
      DO 360 I=1,N
      DO 360 J=1,N
      IF(I.EQ.J) GO TO 360
      IF((KEXT(I).EQ.1).OR.(KEXT(J).EQ.0)) GO TO 360
      IF(SS(I,J).LT.1.0E-20) GO TO 360
      WRITE(3,25) NODE(I),NODE(J),DLONG,DRAD,HSURE,SS(I,J),POWER
      WRITE(5,25) NODE(I),NODE(J),DLONG,DRAD,HSURE,SS(I,J),POWER
  360 CONTINUE
      WRITE(3,26) NEND
      WRITE(5,26) NEND
  400 IF(NEXT.NE.0) GO TO 100
      CALL EXIT
      FND
      SUBROUTINE MATINV(A,N,B,M,DETERM)
C     F1 NBSB MATINV
C     MATRIX INVERSION WITH ACCOMPANYING SOLUTION OF LINEAR EQUATIONS
C
C
CDIMENSIONS FOR MATINV ARE IPIVOT(N),A(N,N),B(N,1),INDEX(N,2),PIVOT(N).
C  N IS THE MAXIMUM VALUE FOR N DEGRE.
      DIMENSION IPIVOT(20), A(20,20), B(20,1), INDEX(20,2), PIVOT(20)
      EQUIVALENCE (IROW,JROW), (ICOLUM,JCOLUM), (AMAX, T, SWAP)
C
C     INITIALIZATION
C
   10 DETERM=1.0
   15 DO 20 J=1,N
   20 IPIVOT(J)=0
   30 DO 550 I=1,N
C
C     SEARCH FOR PIVOT ELEMENT
C
   40 AMAX=0.0
   45 DO 105 J=1,N
   50 IF (IPIVOT(J) 1) 60, 105, 60
   60 DO 100 K=1,N
   70 IF (IPIVOT(K)-1) 80, 100, 740
   80 IF (ABSF(AMAX)-ABSF(A(J,K))) 85, 100, 100
   85 IROW=J
   90 IC0I UM=K
   95 AMAX=A(J,K)
  100 CONTINIIF
  105 CONTINUE
  110 IPIVOT(ICOLUM)=IPIVOT(ICOLUM)+1
C
C     INTERCHANGE ROWS TO PUT PIVOT ELEMENT ON DIAGONAL
C
  130 IF (IROW-ICOLUM) 140, 260, 140
  140 DETERM=-DETERM
  150 DO 200 L=1,N
  160 SWAP-A(IROW,L)
  170 A(IROW,L)=A(ICOLUM,L)
```

```
  200 A(ICOLUM,L)=SWAP
  205 IF(M) 260, 260, 210
  210 DO 250 L=1, M
  220 SWAP=B(IROW,L)
  230 B(IROW,L)=B(ICOLUM,L)
  250 B(ICOLUM,L)=SWAP
  260 INDEX(I,1)=IROW
  270 INDEX(I,2)=ICOLUM
  310 PIVOT(I)=A(ICOLUM,ICOLUM)
  320 DETERM=DETERM*PIVOT(I)
C
C      DIVIDE PIVOT ROW BY PIVOT ELEMENT
C
  330 A(ICOLUM,ICOLUM)=1.0
  340 DO 350 L=1,N
  350 A(ICOLUM,L)=A(ICOLUM,L)/PIVOT(I)
  355 IF(M) 380, 380, 360
  360 DO 370 L=1,M
  370 B(ICOLUM,L)=B(ICOLUM,L)/PIVOT(I)
C
C      REDUCE NON-PIVOT ROWS
C
  380 DO 550 L1=1,N
  390 IF(L1-ICOLUM) 400, 550, 400
  400 T=A(L1,ICOLUM)
  420 A(L1,ICOLUM)=0.0
  430 DO 450 L=1,N
  450 A(L1,L)=A(L1,L)-A(ICOLUM,L)*T
  455 IF(M) 550, 550, 460
  460 DO 500 L=1,M
  500 B(L1,L)=B(L1,L)-B(ICOLUM,L)*T
  550 CONTINUE
C
C      INTERCHANGE COLUMNS
C
  600 DO 710 I=1,N
  610 L=N+1-I
  620 IF (INDEX(L,1)-INDEX(L,2)) 630, 710, 630
  630 JROW=INDEX(L,1)
  640 JCOLUM=INDEX(L,2)
  650 DO 705 K=1,N
  660 SWAP=A(K,JROW)
  670 A(K,JROW)=A(K,JCOLUM)
  700 A(K,JCOLUM)=SWAP
  705 CONTINUE
  710 CONTINUE
  740 RETURN
       END
      SUBROUTINE ORDER
C
C---THIS SUBROUTINE READS THE TOTAL INTERCHANGE AREAS AS A LIST (AS
C   GENERATED BY CNVUFAC) AND GENERATES THE LOWER TRIANGULAR MATRIX.
C   ELIMINATE ALL HEADING CARDS AND END WITH A BLANK CARD.0
C   NOTE THAT INTERCHANGE AREAS NOT PRESENT ARE ASSUMED TO BE ZERO.
C   DUPLICATE INTERCHANGE AREA DATA ARE AVERAGED.
C
C---NOTE THAT DIMENSIONING IS CURRENTLY FOR 20 NODES.  DIMENSIONS IN
C   SUBROUTINE MATINV MUST BE INCREASED IF DIMENSIONS IN MAIN
C   PROGRAM ARE INCREASED TO HANDLE MORE THAN 20 NODES.
      COMMON HEAD(8),A(20),E(20),SSB(20,20),SS(20,20),F(20,20),
```

```
      1 RHO(20),B(20),WE(20,20),KE(20),NODE(20),KEXT(20),FNAME(3),N
       DIMENSION KTR(20,20)
C
C---FORMAT STATEMENTS.
    1 FORMAT(2I10,E10.3)
    2 FORMAT(56H A DUPLICATE NODE IDENTIFICATION HAS BEEN FOUND FOR NODE
     1 ,I10)
    3 FORMAT(/85H A NODE NUMBER ON THE FOLLOWING CARD WAS NOT LISTED ON
     1THE NODE IDENTIFICATION CARDS./2I10,E10.3)
C
C---INITIALIZE. AND CHECK FOR DUPLICATE NODE NUMBER IDENTIFICATION.
      DO 100 I=1,N
      DO 100 J=1,I
      IF(I.EQ.J) GO TO 90
      IF(NODE(J).NE.NODE(I)) GO TO 90
      WRITE(3,2) NODE(I)
      CALL EXIT
   90 SSB(I,J)=0.0
      KTR(I,J)-0
  100 CONTINUE
C
C---INPUT AND FORM LOWER TRIANGULAR SSB MATRIX.
C    NODE1, NODE2=NODE NUMBERS (NOTE THAT THE ORDER OF THESE 2 NODES
C    MAKES NO DIFFERENCE).
C    SSBB=A*FB=BLACK BODY TOTAL INTERCHANGE AREA.
  105 READ(2,1) NODE1,NODE2,SSBB
      NODE1=IABS(NODE1)
      NODE2=IABS(NODE2)
      IF((NODE1.NE.0).OR.(NODE2.NE.0).OR.(SSBB.NE.0.0)) GO TO 108
      DO 107 I=1,N
      DO 107 J=1,I
      IF(KTR(I,J).LT.1) KTR(I,J)=1
      SSB(I,J)=SSB(I,J)/FLOAT(KTR(I,J))
  107 CONTINUE
      RETURN
  108 ICHK=0
      JCHK=0
      DO 150 IJ=1,N
      IF(NODE1.NE.NODE(IJ)) GO TO 110
      II=IJ
      ICHK=1
  110 IF(NODE2.NE.NODE(IJ)) GO TO 120
      JJ=IJ
      JCHK=1
  120 IF((ICHK.EQ.1).AND.(JCHK.EQ.1)) GO TO 130
      IF(IJ.LT.N) GO TO 150
      WRITE(3,2) NODE1,NODE2,SSBB
      CALL EXIT
  130 I=II
      J=JJ
      IF(I.GE.J) GO TO 140
      I=JJ
      J=II
  140 SSB(I,J)=SSB(I,J)+SSBB
      KTR(I,J)=KTR(I,J)+1
      GO TO 105
  150 CONTINUE
      GO TO 105
      END
*....................FILE OPEN SUBROUTINE.............1/17/75....
```

```
      SUBROUTINE OPENFN(FN,NT,LG,IE,MWX)
*
*         24 SEP 74 VERSION     D. LAI
*
*         INPUT  FN : NAME OF FILE TO BE OPENED OR CREATED
*                NT : TAPE NUMBER TO BE ASSIGNED TO FILE
*                     = 0 : NO TAPE NUMBER TO BE ASSIGNED
*                LG : LENGTH OF FILE IF CREATED
*                IE : OPTIONS AVAILABLE -
*                     =] -1 : OPEN AND MAKE READ-WRITE
*                     =  -2 : OPEN ONLY
*                     =  -3 : OPEN OR CREATE READ-WRITE
*                     =  -4 : CREATE ONLY
*                     =  -5 : CREATE OR DESTROY AND CREATE
*        OUTPUT  LG : LENGTH OF FILE OPENED
*                IE : ERROR INDICATOR -
*                     = 0 : SUCCESSFUL EXECUTION
*                     ] 0 : FLOE ERROR NUMBER FOR OPEN INSTRUCTION
*               MWX : MINUS WORD OR IOC WORD RETURNED IF
*                     VARIABLE WAS INCLUDED IN ARGUMENT LIST
*
*...................................................................
      DATA IT/59/
    1 FORMAT(*FILE ALREADY EXISTS*)
    2 FORMAT(*CANNOT CREATE FILE *A10* : ERROR 0*I2)
    3 FORMAT(*CANNOT WRITE FILE *A10)
    4 FORMAT(*CANNOT OPEN FILE *A10)
      CALL ARG(NARG) $ MODE=0
      IF(FN.EQ.55555555555555555555B.OR.FN.EQ.0)GO TO 13
      IF(NT.LE.0)GO TO 5 $ REWIND NT $ CALL ASSIGN(NT,FN)
    5 IF(NARG.GT.3)MODE=-IE $ IF(MODE.LT.1)MODE=1
      LN=0 $ IF(NARG.GT.2)LN=LG
      IF(LN.LE.0)LN=2560 $ LNX=LN $ IF(MODE.LT.4)12,7
    6 MODE=6
    7 LN=LNX $ CALL DEVICE(6HCREATE,FN,LN,NE,MW)
    8 IF(NE-1)15,9,10
    9 IF(MODE.NE.4)GO TO 10 $ IF(NE.EQ.1)WRITE(IT,1)
      IF(NE.GT.1)WRITE(IT,2)FN,NE $ GO TO 16
   10 IF(NE.EQ.1.AND.MODE.EQ.5)GO TO 11 $ WRITE(IT,2)FN,NE $ GO TO 16
   11 CALL DEVICE(4HOPEN,FN,L) $ CALL DEVICE(7HDESTROY,FN) $ GO TO 6
   12 CALL DEVICE(4HOPEN,FN,LN,NE,MW) $ IF(NE.EQ.0)GO TO 15
      IF(NE.EQ.2.AND.MODE.EQ.2)GO TO 15 $ IF(NE.NE.2)GO TO 13
      CALL DEVICE(6HCLOSEW,FN) $ CALL DEVICE(4HOPEN,FN,LN,NE,MW)
      IF(NE.EQ.0)GO TO 15 $ WRITE(IT,3)FN $ GO TO 16
   13 IF(MODE.EQ.3)GO TO 14 $ WRITE(IT,4)FN $ GO TO 16
   14 MODE=5 $ GO TO 7
   15 NE=0
   16 IF(NARG.GT.2)LG=LN $ IF(NARG.GT.3)IE=NE $ IF(NARG.GT.4)MWX=MW
      RETURN $ END
```

Available from : National Technical Information Service - U.S. Department of Commerce
5285 Port Royal Road - Springfield, VA 22161 - $6.00 per copy - (Microfiche $3.50)